
Introduction

All commercial Smalltalk environments were designed with large desktop machines in mind. So tools such as debuggers and browsers keep track of the source in data structures internal to the environment. If we were to build a Smalltalk environment that runs on a small device like a palm-sized computer or a teleoperated robotic device such as a fire deployment probe, we might not want or need these data structures at run-time.  Method categories alone occupy a lot of space.



Of course, Smalltalk already keeps the method source in external files; namely, the sources and change files. If we were to built a headless Smalltalk (the term for a Smalltalk environment that has no windowing facilities) controlled via remote browsers and debuggers, we might want to move as much as we can out of the image itself. That way, we might even be able to simplify the structure of the classes a wee bit. 



In some Smalltalk systems, portions of the source may be hardwired into constant DLLs making it extremely difficult to retrieve without running the actual image. Otherwise, we could build a facility relatively easily that parses the code in the sources and changes files and hands it over to a remote system. We could eliminate the need to parse the information by switching to a simple binary format. It wouldn’t be as convenient for users who regularly grab bits and pieces from these files using a simple text editor, but there are benefits. 



One advantage is that we could store objects that are more interesting than just a bunch of strings. For example, we could design a source object that contains the source, method category, developer identifier, test methods, and documentation that includes text and pictures. This complex object could reside entirely on disk being materialized only when the actual method is selected for display in a browser or debugger. The big advantage is that the image (main memory) would not have to pay for the space requirements; the file system would instead. 



Once we’ve decided on such an extension, there is no need to insist that it only be used for storing complex objects representing methods. In fact, this facility is actually on object storage facility. To emphasize this role, I’ll simply call it a data manager.

Designing The Data Manager

A data manager as described above is just a simple data base; an alternative view is that it is nothing but a glorified dictionary. Presumably, we supply it with the keys and the data to save away and it allows us to retrieve and modify the data given the key. We could store arbitrary information with a class using its unique class name as the key. We could use the class name and selector name pair as a key to store method source objects. 



<PULL QUOTE from above: A data manager is nothing but a glorified dictionary.>



There is one requirement that we have to pay attention to when designing such a data manager. The information residing in the external file must remain in a consistent state even if a user crashes. We can explicitly design the data manager to remain consistent even if it runs out of space. Of course, in that situation, it’s not possible to actually store the new data. But that’s the most that we should lose. We should not, for example, lose everything that was done in an entire programming session.



<PULL QUOTE from above: The external file must remain in a consistent state even if a user crashes.>



Even if we can satisfy the above requirement, there is one that we will not be able to deal with. Input/output errors due to hardware problems cannot be eliminated. If the file system itself is not able to copy a defective file, then neither can we copy it nor can we correct it.



To deal with the fundamental problem, we just have to make sure that file operations requiring additional space detect the failure. We can do this with standard exception handling mechanisms. For example, in the Smalltalk environment we are using, we could execute



	[… execute file operations …] 

		on: Error do: [:exception | 

			… react to error condition …]



We don’t need to be this careful for all file operations. Operations that merely overwrite existing file areas must succeed (barring unrecoverable input/output errors). Such error are normally reported even if we do nothing.

	

We designed our first data manager to open by retrieving/constructing a dictionary of key/file-position values. The object values were, of course, retrievable on demand and so remained on disk. To support incremental additions (we did not want to have to create an entirely new file each time we used it), we designed it so that delta dictionaries could be appended.  By a delta dictionary, we mean one that contains only the new keys (or modified ones) that are inserted during use. 



We didn’t want to store one-element delta dictionaries, so we assumed a reasonably large but fixed size buffer area, say 10k bytes, for delta dictionaries. Once we allocated such a buffer at the end of a file, new key-value insertions required the values to be placed after this pre-allocated space. Clearly, we couldn’t wait until the delta dictionary was full before storing it out on disk because anyone could accidentally pull out the power plug at any time. We needed to output the delta dictionary either incrementally which is messy or in totality each time an addition was made. The latter would be to slow if we were designing an arbitrary data base but acceptable for an application where it occurs only when a user saves a method. Even so, how do we tell when the delta dictionary becomes too big for its buffer area. Keep in mind that we decided to use arbitrary objects for values and (why not) arbitrary values for the keys too. And how much space is needed for a dictionary anyway! We decided to output the delta dictionary in an internal stream first thus giving us the ability to predict how much space was needed. If the buffer space limits were exceeded, we needed to create a new buffer, and yet another delta dictionary with just one entry (the last one), and deal with the issue of linking these delta dictionaries together on disk.



Ultimately, we had a 3-class design that did the job, but we felt that it was too complex. Ultimately, we started over and repeated a simple mantra “If it behaves like a dictionary, it is a dictionary…. If it …” The following is the result.

The Data Manager Class



The DataManager class (see Listing 1) was designed to maintain (1) a dictionary of key/file-position pairs, (2) a stream opened on the file containing the data manager information, (3) the position in the stream of the information related to the last dictionary entry (this way, we can provide a linked list of entries), and (4) a workStream. Since the keys and values are arbitrary, we use the object filer to store objects into the file; it can handle complex circular objects. 



We actually output a dictionary entry (a key-value pair) in 3 pieces in the following order: (1) last-entry-location, (2) key-object, and (3) value-object. Thus we maintain a reverse order linked list (linked last to first). We are considering changing it to a forward linked list but we’ll make that small change later. The linked list allows us to read in the keys and note the location of the values (but not read them in). Because the linked list is backward linked, we have to store the location of the last entry somewhere. 



Since we can’t easily work backward from object filer output, we simply stored the position of the last entry at the beginning of the file (again in object filer format). We could have used a non-object filer format but that would have implied an unnecessary assumption; i.e., that the value is a small integer and therefore fits in 4 bytes, or that it might be slightly larger in which case we would need a string-like representation to encode either a small integer or a large positive integer. 



When a data manager file is opened, we read in all the entries in reverse order of insertion (last to first). We actually pick up only the key/value-file-position pair for storage into our private dictionary; the value itself which is typically a very large object remains in the file. We only update the value in the dictionary if the key is new (since the most recent value associated with the key is in fact the last one).



To verify that the implementation actually does what we described above, consider tracking the methods invoked when executing something like



	| manager |

	manager := DataManager new.

	manager openOn: 'managerFile'.

	manager at: #key put: (Set with: 'anything').

	manager close.



The openOn: method either executes initializeNew or initializeOld depending on whether or not the file exists. If it does (the more complex case), the linked list of key/value-file-position information is input in reverse order. If the key already exists, it is not replaced.



The at:put: method makes use of a private version that returns whether or not space was successfully allocated. This is not used by the at:put: method but it is used by the compress method to terminate early. 

Method privateAt:put: uses “growBy: amount ifSucceeded: block1 ifFailed: block2” to detect whether or not adequate space is available before writing into this newly obtained space. So that we can actually determine how much space we need, we first output the last position, the key, and the value into the workspace stream. However, we do it in two pieces: the last position and key in one chunk. We retrieve it in string1. Then the value object is output in another chunk which we retrieve as string2. We now know how much space to get via the special growBy:… method mentioned above.



If there is sufficient space, we output string1 which contains the last position and the key into the external file. The next position in the external file can then be associated with the key in our internal dictionary. We then output string2 which contains the value object. Once this is done, we execute finalize which records the position of this last entry at the beginning of the file. It also flushes the file ensuring that the external file contains all current information.



Note that the grow method makes use of a special method successfullyExecuted: which executes its block under control of an exception handler which catches all errors. Since successfullyExecuted: is only used when we grow the size of the file (by repositioning by a sufficient amount beyond the end of the file, outputting a single irrelevant character, and then flushing the file), we can be sure that the error will be an “out of space” error. If no error results, execution continues as discussed above.



If we fail to obtain sufficient space, we provide the user with three options: (1) simply delete old files, make room, and try again, (2) give up which causes a simple error notifier to appear, or (3) cancel which causes the original at:put: to do nothing (but nevertheless continue). 



<PULL QUOTE from above: If we fail to obtain sufficient space, we provide the user with three options.>



Note: Although a key is associated only with a value object’s file position in our private dictionary, a user querying the data manager via at:, at:ifAbsent:, or keysAndValuesDo: actually gets the value object since it is explicitly retrieved for the user.

Conclusions

We implemented a data manager that could be used to replace the current source management facility. It’s advantage is that it can incorporate arbitrary objects, not just method source. Moreover, the information can be easily retrieved by remote browsers and debuggers. 



We have not actually tried to replace an existing source manager or tried to generalize on the information that could be associated with a method or a class for that matter. 



It’s also embarrassingly simple but we must admit that we didn’t start out being simple. A friendly slap woke us up to this simpler solution.





Listing 1 The external data manager class.



class:								ExternalDataManager

superclass:						Object 

instance variables:		dictionary stream last workStream

class variables:				Debugging



class methods

examples

example1

	"Fragments useful for testing."

	Junk := ExternalDataManager new.

	Junk openOn: 'junk.mng'.

	Junk close.



	Junk at: 1 put: 'Testing'.

	Junk at: 1.



	#('key1' 'key2' 'key3' 'key4' 'key5') do: [:key |

		| string |

		string := ''.

		100 timesRepeat: [string := string, key, ' '].

		Junk at: key put: string].



	ExternalDataManager debugging: true.

	ExternalDataManager debugging: false.



	#('key6' 'key2' 'key3' 'key7' 'key5') do: [:key |

		Junk at: key put: key].



    Junk keysAndValuesDo: [:key :value |

        MessageBox 

			message: key printString, 

				' has value ', value printString].



	Junk at: 'key8' put: 'Try this with debugging true'.

	Junk at: 'key8' ifAbsent: ['Not there'].



	Junk compress.



	Smalltalk removeKey: #Junk



querying

headerSize

	^40 "bytes"



debugging: aBoolean

	Debugging := aBoolean

debugging

	^Debugging == true "Works even if not initialized"



instance methods

private initializing/finalizing

initializeNew

	self growBy: self class headerSize.

	last := 0.

    self finalize



initializeOld

	"Previously constructed map."

	| key value workingLast valuePosition |

	stream isEmpty ifTrue: [

		self error: 'Not an external data manager file'].

	stream position: 0.

	last := workingLast := self load.

	[workingLast = 0] whileFalse: [

		stream position: workingLast.

		workingLast:= self load. "previous one"

		key := self load.

		valuePosition := stream position. "but don’t read"

		(dictionary includesKey: key) ifFalse: [

			dictionary at: key put: valuePosition]]



finalize

	stream position: 0.

	ObjectFiler dump: last on: stream.

	stream position: stream size.



private opening/closing support

alreadyClosed

	^stream isNil

  

alreadyOpen

	"Warns user and asks for directions. Returns true if already open."

	| reply name |

	self alreadyClosed ifTrue: [^false].

	name := stream file pathName.

	reply := MessageBox

		threeStateNotify: 'File "', name, '" already open'

		withText: 'Close and continue?'.

	reply == true ifTrue: [self close. ^false].

	reply == false ifTrue: [

		self error: 'file "', name, '" already open'].

	"Otherwise, it's nil."

	^true



public opening/closing

openOn: fileName

	"Open a stream and read in data."

	| name |

	self alreadyOpen ifTrue: [^self].

	dictionary := Dictionary new.

	workStream := '' asStream.

	name := File fullPathName: fileName.

	(File exists: name)

		ifTrue: [

			stream := File pathName: name.

			self initializeOld]

		ifFalse: [

			stream := File newFile: name.

			self initializeNew].



close

	stream isNil ifTrue: [^self]. "already closed"

	stream close.

	stream := nil.



public accessing/modifying

at: key ifAbsent: aBlock

	"The value is in external memory. The private 

	dictionary keeps track of its file position."

	| where |

	where := dictionary at: key ifAbsent: [^aBlock value].

	stream position: where.

	^ObjectFiler loadFrom: stream.

 

at: key

	^self 

		at: key 

		ifAbsent: [

			self error: 'key ', key printString, ' does not exist']



at: key put: value

    self privateAt: key put: value.

    ^value "maintain dictionary semantics"



keysAndValuesDo: aBlock

    dictionary keys do: [:key |

        aBlock value: key value: (self at: key)]



private accessing/modifying

privateAt: key put: value

    "Returns whether or not entry was successfully inserted.

	Careful: dictionary contains key and stream position of value. Stream contains position of previous entry, key, and then value."

    | string1 string2 valuePosition |

    workStream reset; truncate.

    self dump: last; dump: key.

    string1 := workStream contents.

    workStream reset; truncate.

    self dump: value.

    string2 := workStream contents.

    self growBy: string1 size + string2 size

		ifSucceeded:  [:newLast |

            stream

                position: newLast;

                nextPutAll: string1.

            valuePosition := stream position.

            stream

                nextPutAll: string2.

            dictionary at: key put: valuePosition.

            last := newLast.

            self finalize.

			^true]

		ifFailed:  [^false "Don't append anything."]



private growing/loading/dumping

growBy: amount

	"Make sure user gets a message if there is no space."

	self growBy: amount ifSucceeded: [:newLast | ] ifFailed: []



growBy: amount ifSucceeded: block1 ifFailed: block2

	"Go to the end of stream, output anything, and force to disk."

	| newLast |

	newLast := stream size.

	(self successfullyExecuted: [

		"Next line should be removed as soon as

		debugging is no longer needed."

		self class debugging ifTrue: [self smile].

		stream

			position: newLast + amount - 1;

			nextPut: $a; "anything at all"

			flush])

		ifTrue: [block1 value: newLast]

		ifFalse: block2

   

dump: object

	"Careful: dumps only on the work stream."

	ObjectFiler dump: object on: workStream



load

	"Careful: Not the opposite of dump. Loads from

	the external stream, not the work stream."

	^ObjectFiler loadFrom: stream



private metalevel facilities

successfullyExecuted: aBlock

	"Only used when attempting to get NEW space."

	| reply |

	aBlock on: Error do: [:e |

		reply := MessageBox

			threeStateNotify: 'Out of space'

			withText: 'Delete unnecessary files and try again?'.

		reply == true ifTrue: [^self successfullyExecuted: aBlock].

		reply == false ifTrue: [self error: 'out of space'].

		"Otherwise, it's nil."

		^false].

	^true



public querying

fileName

    self alreadyClosed ifTrue: [^''].

    ^stream file pathName



fileNamePrefix

    | name |

    name := self fileName.

    ^name 

		copyFrom: 1 

		to: (File indexOfFilenameIn: name) - 1



fileNameSuffix

    | name |

    name := self fileName.

    ^name 

		copyFrom: (File indexOfFilenameIn: name) 

		to: name size



public compressing

compress

    "This operation is intended to be safe."

    | manager oldName newName |

    (MessageBox confirm: 'Begin Compress') ifFalse: [^self].

    oldName := self fileName.

    newName := self fileNamePrefix, '~$', self fileNameSuffix.

    manager := self class new.

    manager openOn: newName.

    self keysAndValuesDo: [:key :value |

        (manager privateAt: key put: value) ifFalse: [

            MessageBox 

				message: 'Terminated unsuccessfully. ',

					'Original unchanged'.

            manager close.

            File remove: newName.

            ^self]].

    manager close. self close.

    File remove: oldName.

    File rename: newName to: oldName.

    self openOn: oldName.

    MessageBox message: 'Terminated successfully'

